

Advanced SAS Graph Template Language (GTL) with Practical Examples from Oncology Trials

Ceng Qian, Xu Wen, Wei Lei, and Ling Han, Gilead Sciences, Inc.

ABSTRACT

With increasing complexity of clinical trials, the business needs for high-dimensional and dynamic data visualization become more crucial to help decision-making and drug development. Compared to traditional SAS SG procedures, SAS Graph Template Language (GTL) provides more user-defined features and flexibility. In this paper, we illustrated several practical examples of customized figures using advanced SAS GTL code, including multi-layer waterfall plots, data-driven heatmaps, longitudinal swimmer plots, and cycle-based adverse event (AE) incidence plots from Oncology trials. The purpose of this paper is to motivate SAS programmers to utilize the powerful GTL and create high quality graphs.

INTRODUCTION

Oncology trials have diverse patient-level data, including longitudinal efficacy assessments, treatment exposure, safety events, baseline disease characteristics, and biomarker measurements. Traditional outputs and isolated figures often limit the ability to assess relationships across these domains, particularly when evaluating dose-response trends and temporal associations between efficacy and safety.

SAS Graph Template Language (GTL) provides the flexibility required to construct highly customized, multi-panel clinical graphics while maintaining control over layout, attributes, and scalability. This paper presents a visualization framework developed using SAS GTL to support exploratory analysis in oncology trials, with an emphasis on programming design principles and reusable implementation strategies.

SAS GTL generates figures by defining a STATGRAPH template in PROC TEMPLATE, placing plots inside multiple LAYOUT OVERLAY ... ENDLAYOUT containers, and then rendering the template with PROC SGRENDER using a data set. The LAYOUT OVERLAY block is the single-cell canvas that combines compatible 2-D plot statements (e.g., SCATTERPLOT, SERIESPLOT, BARCHART), plus axes, legends, and reference lines; PROC SGRENDER binds the template to the data to produce the graph.

MULTI-DIMENSIONAL WATERFALL AND HEATMAP PLOTS

The multidimensional waterfall plot of tumor burden and clinical relevance is utilized to visualize individual patient responses across treatment arms or other classification groups. The plot enabled the study team to assess associations between tumor shrinkage, best overall response, treatment exposure, participant study duration, and key disease characteristics (e.g. biomarker levels). The visual summaries facilitated rapid identification of response patterns and outliers.




Figure 1. Waterfall Plot of Overall Response over Study Duration and Maximum Reduction from Baseline in Target Lesion by Best Overall Response

STEP-BY-STEP PROCEDURE

Step 1. Define Objectives

Determine what clinical endpoints and variables need to be visualized.

The figure is divided into three components:

1. **Study duration**, with longitudinal tumor response segmented by color over time.
2. **Maximum tumor reduction from baseline**, presented as a waterfall plot and colored by best overall response.
3. **Baseline biomarker values**, displayed as aligned heatmap.

Detailed plot specifications are illustrated in Figure 2.




Figure 2. plot specifications break down.

Step 2. Design Plot Data

We can conceptualize the layout from the x-axis to the y-axis. The x-axis represents a fixed variable – ID ordered by maximum tumor reduction from baseline. In contrast, the y-axis incorporates multiple variables, such as study duration and response, treatment status and survival status, maximum reduction in target lesion from baseline and best overall response, treatment assignment, and biomarker information. This section outlines the principles guiding the dataset architecture.

2.1 Unifying Multiple ADaM Sources Into a Single Plot Dataset

The figure integrates data from ADSL, ADRS, ADTR/ADEFF, and biomarker datasets, and we structured them using a BDS (Base Data Structure) format.

All visual elements, bars, markers, events, heatmap tiles, labels, are stored as rows within the same dataset, distinguished by classification variables such as:

- PARAM (element type: Response, Treatment Status, Percent Change, Biomarker, etc.)
- PARCAT / PARCAT2 (panel or sub-panel grouping)
- ID (plot x-axis ordering)
- Y_ variables defining the vertical placement of each component

This structure allows GTL to read one dataset and layer elements in the correct order without additional reshaping.

2.2 Creating a Stable X-Axis Ordering Variable

Because the visualization compares participants across multiple clinical domains, the dataset contains a single, stable ordering variable ID.

2.3 Vertical Structuring Using Y-Coordinate Bands

To avoid conflicts between heterogeneous components (bars vs. heatmaps vs. markers), the dataset assigns pre-defined y-bands, each corresponding to a conceptual layer. Examples include:

- AWK_Y – response-by-week stacked bars
- TRTDURW_Y – treatment duration bar
- STATUS_Y, DTH_Y – event markers
- PCHG_Y – waterfall bar height
- TRT_Y – treatment label position
- B1_Y, B2_Y, B3_Y, B4_Y – biomarker lane anchors

These y-coordinates serve as layout infrastructure enabling GTL to position multiple elements cleanly on the same plot.

2.4 Standardizing Clinical Categories for Plot Consistency

To ensure reproducible mapping to colors, shapes, and fill patterns, the dataset must encode clinical categories in standardized variables, for example:

- RESPN, BORN –response categories
- STATUS, DTH – treatment status categories, death indicator

Standardization ensures that discrete attribute maps in GTL can reliably interpret categories without ambiguity.

The data structure is also illustrated in the table below (Table 1).

Figure components	“PARAM”	Y-axis	Source	Grouping Variable	Structure	“PARCAT”
Study Duration	“Response”	Day of Response assessment	ADRS	ADRS.OVRLRESP (CR/PR/SD/PD)	Per participant Per RS timepoint	PARCAT2
	“Treatment Status”	EOTDY	ADSL	ADSL.COMPTL (Ongoing/AE/PD/Other)	Per participant	PARCAT1
	“Death”	DTHDY	ADSL	/		
Maximum Reduction	“Percent Change”	PCHG	ADTR	ADEFF.BOR (CR/PR/SD/PD)	Per participant	
	“Treatment”	Min PCHG – a (self defined position)	ADSL	TRT01A		
Biomarker	“Biomarker”	XXXX (self defined position)	Biomarker	/		

Table 1. Plot data structure

Step 3. GTL Template

We implemented a custom GTL graph using PROC TEMPLATE to visualize treatment timelines, response, and biomarker dynamics in a single, coordinated figure.

The template defines discrete attribute maps for response categories (Pre-assessment, CR, PR, SD, PD, NE) and status events (Ongoing, AE, PD, Other, Death) to ensure consistent color fills and marker symbols across panels. Four continuous range attribute maps encode biomarker intensity on a white-to-color gradient for Biomarkers 1–4, which are rendered as heatmap bands annotated with bold labels.

The figure employs a 2x2 lattice:

- The upper-left overlay displays study duration (weeks) on a linear y-axis with stacked bars for response by week, thin black bars for treatment exposure, and scatter markers for status and death events.
- The upper-right overlay contains legends for treatment status.
- The lower-left overlay shows waterfall-style bars of maximum percent change from baseline, grouped by best overall response and complemented with fill-patterned bars to distinguish unconfirmed or censored categories; two horizontal reference lines at -30% and +20% delimit RECIST-like thresholds.
- The lower-left overlay also shows heat maps of baseline biomarker values.
- The lower-right overlay is a supplementary single-column lattice to the right provides compact legends for response and biomarker gradients, with titled sections for each biomarker's continuous legend.
- Axis ranges, tick sequences, and layout weights are controlled to harmonize scales and emphasize time-course (top) and magnitude-of-change (bottom) views, with clipping and borders disabled for a clean presentation.

Patial GTL template code is provided below for illustration purpose (program 1)

```

/*=====
Program: Waterfall + Timelines + Biomarker Heatmap (GTL)
Template: STATGRAPH WTF (stored in WORK.TEMPLAT)
Purpose :
  - Top-left : Study duration (weeks), treatment exposure, status & death
  - Top-right : Treatment/participant-status legend
  - Bottom-left: Waterfall of % change from baseline + biomarker heatmaps
  - Bottom-right: Legends for response and biomarker gradients
Dependencies (provided by plot-ready dataset):
  - X ordering: ID (stable participant rank across all panels)
  - Y scaffolds: AWK_Y, TRTDURW_Y, STATUS_Y, DTH_Y, PCHG_Y, TRT_Y,
                 B1_Y-B4_Y (biomarker lane anchors)
  - Encodings: RESPN (response), BORN (BOR), STATUS, DTH,
                 B1TN-B4TN (scaled biomarker intensities)
Macros used for axes and lane heights:
  &MAXWK, &VIEWMIN, &MAXPCHG, &YBOUND1-&YBOUND8
=====*/

proc template;
/* Canvas for the entire figure */
define statgraph wtf / store=work.templat;

  /*=====
     Section A. DISCRETE attribute maps for categorical encodes
     - RESPN : time-segment response colors (top-left)
     - BORN : waterfall BAR colors by BOR (bottom-left)
     - STATUS: status markers (AE/PD/Other/Ongoing)
     - DTH : death marker
  =====*/

  beginngraph / designheight=11in designwidth=6in border=false;
    discreteattrmap name="resp" / discretelegendentrypolicy=attrmap;
      value 'Pre-assessment' / fillattrs=(color=lightgrey );
      value 'CR' / fillattrs=(color=orange);

```

```

        value 'PR' / fillattrs=(color=cx1a9641);
        value 'SD' / fillattrs=(color=cx3d5aae);
        value 'PD' / fillattrs=(color=cxd7191c);
        value 'NE' / fillattrs=(color=cxffffbf);
    enddiscreteattrmap;
    discreteattrvar attrvar=respn var=respn attrmap="resp";

    (Other similar discrete attributes are omitted here.)

/*=====
Section B. RANGE attribute maps for continuous biomarkers
- Each biomarker gets its own white-color gradient
- *_TN variables are scaled intensity inputs
=====*/

    rangeattrmap name="blmap";
        range 0 - max / rangecolormodel=(white cxaf4b91);
    endrangeattrmap;

    (Other similar range attributes are omitted here.)

/*=====
Section C. Primary 2x2 lattice
- Top-left : timelines panel (study duration, exposure, status)
- Top-right : treatment status legend
- Bottom-left: waterfall + biomarker heatmaps
- Bottom-right: legends column
=====*/

layout lattice / columns=2 rows=2 columnweights=(0.8 0.2)
                rowweights=(0.3 0.7) rowdatarange=union border=false;

/*-----
(1) Top-left overlay: Study Duration, Exposure, Events
- Y axis in weeks using &MAXWK with 6-week ticks
- Stacked bars by RESPN show time-segment response color
- Line indicates exposure
- Status/death markers overlay the same panel
-----*/

    layout overlay / border=false walldisplay=none
        yaxisopts=(type=linear label="Study Duration (Weeks)" griddisplay=on
            linearopts=(viewmin=0 viewmax=&maxwk
                tickvaluesequence=(start=0 end=&maxwk increment=6)))
        xaxisopts=(type=discrete display = none);

        /* Segmented weeks colored by response category */
        Barchart x=id y=awk_y / Group=respn display=(FILL)
            groupdisplay=stack name="resp";
        /* Status and death event markers (placed at STATUS_Y / DTH_Y) */
        scatterplot x=id y=status_y / group=status name="status";
        scatterplot x=id y=dth_y / group=dth name="dth";
        /* Line (thin bar) for treatment exposure duration (in weeks) */
        Barchart x=id y=trtdurw_y / barwidth=0.05 fillattrs=(color=black);

    endlayout;

/*-----
(2) Top-right overlay: legend for status/death
-----*/

    layout overlay / border=false;
        DiscreteLegend "status" "dth" / title="Treatment Status"
            halign=left valign=center across=1
            border=false displayclipped = true;
    endlayout;

```

```

/*-----
(3) Bottom-left overlay: Waterfall + Biomarker Heatmaps
- Y axis shows % change with RECIST-like refs at -30% / +20%
- Waterfall bars colored by BOR (BORN)
- Patterned variants for confirmed/unconfirmed categories
- Treatment label under each bar
- Four biomarker heatmap lanes + lane labels
-----*/

layout overlay / border=false walldisplay=none
  yaxisopts=(type=linear label="Maximum Reduction from Baseline (%)"
    labelattrs=(size=9pt) griddisplay=on
    linearopts= (tickvaluelist=(-100 -50 -30 0 20 50 100 150)
    viewmin=&viewmin viewmax = &maxpchg))
  xaxisopts=(type=discrete display = none);

  /* Main waterfall by BOR */
  barchart x=id y=pchg_y / group=born display=(FILL) name="bor";
  /* Hatch fills to differentiate confirmatory states */
  barchart x=id y=pchg_yuc / group=borucc Display=(FillPattern)
    fillpatternattrs=(Color=CXA9A9A9 pattern=x1)
    groupdisplay=cluster name="boruc";

  /* Treatment assignment label rotated under each participant */
  textplot x=id y= trt_y text=trt_c /
    textattrs=(size=7.5pt) rotate=90;

  /* Biomarker 1-4 lane + label (bounded vertical band) */
  heatmapparm x=id y=bl_y colorresponse=bltn /
    discretex=true name="blmap" yboundaries=(&ybound1 &ybound2);
  textplot x=id y=bl_y text=blt / textattrs=( size=6pt weight=bold) ;

  (Other similar statements are omitted here.)

  /* RECIST-like horizontal references for quick thresholds */
  referenceline y=20 / lineattrs=(pattern=MediumDash thickness=1 color=black);
  referenceline y=-30 / lineattrs=(pattern=MediumDash thickness=1 color=black);
endlayout;

/*-----
(4) Bottom-right column: Legends (response + biomarkers)
- Multi-row lattice to stack titles + continuous legends
-----*/

layout lattice / columns=1 rows=9
  rowweights=(0.628 0.018 0.075 0.018 0.075 0.018 0.075 0.018 0.075)
  columndatarange=union border=false;

  /* Response legend (BOR + pattern variants) */
  layout overlay / border=false pad=(top=1in) ;
  DiscreteLegend "bor" "boruc" "borc" / title="Response"
  titleattrs=(size=9pt) valueattrs=(size=8pt)
  halign=left valign=top across=1 border=false displayclipped=true;
endlayout;

  /* Biomarker 1-4 title + color gradient legend */
  layout overlay / border=false ;
  drawtext textattrs=(size=6pt weight=bold) "Biomarker 1" ;
endlayout;

  layout overlay / border=false ;
  continuouslegend "blmap" / valueattrs=(size=4pt) orient=horizontal
  valign=top border=false;
endlayout;

  (Other similar layouts are omitted here.)

```

```

endlayout; /* end bottom-right legends column */
endlayout; /* end 2x2 primary lattice */
endgraph;
end;
run;

```

Program 1. GTL template for waterfall plot

Step 4. Render the figure using the prepared plot data and GTL template

After creating the plot data and defining the GTL template, we use the SGRENDER procedure to generate the final figure.

```

proc sgrender data=plotdata template= wtf; run;

```

A VARIATION IN TUMOR REDUCTION ACROSS TREATMENT GROUPS

Here is an alternative version of the waterfall plot displays maximum tumor reduction with color coding by treatment group (Figure 3). We used similar approach to generate the figure.




Figure 3. Waterfall Plot of Overall Response over Study Duration and Maximum Reduction from Baseline in Target Lesion by Treatment Group

SWIMMER LANE PLOTS

COMPARATIVE SWIMMER LANE ANALYSIS OF TREATMENT EXPOSURE AND RESPONSE (FIGURE 4)

Constructed to visualize individual patient exposure timelines across both prior therapies and the current investigational product. These plots integrate dosing schedules, treatment duration, patient responses, and relevant clinical events, enabling direct comparison of therapeutic impact.




Figure 4. Swimmer Plot of Dosing, Objective Response and Treatment Status

A custom SAS GTL STATGRAPH template creates a two-row lattice:

- The main panel shows participant timelines by Study Day, and the lower panel displays legends. DISCRETEATTRMAP/DISCRETEATTRVAR pairs define consistent markers for treatment status, dose, overall response, and BOR. The overlay uses SERIESPLOT for treatment and concomitant medication intervals, and SCATTERPLOT for events such as doses, responses, prior therapy, and death. Axes are labeled “Participant ID” and “Study Day,” with gridlines and a reference line at Day 1.
- Legends for treatment, response, prior therapy, status, and death are arranged in the second row for clarity.

Patial GTL template code for swimmer plot Figure 4 is shown below (Program 2):

```

/*=====
Program: Swimmer Plot - Dosing, Response, Prior Therapy & Status (GTL)
Template: STATGRAPH SWIM (stored in WORK.TEMPLAT)
Purpose:
  - Main panel: participant timelines (Study Day) with exposure, doses,
    responses (overall response and BOR), prior therapy markers,
    treatment status, and death.
  - Second row: compact legends for dose, response, prior therapy, status,
    and participant status.
Plot-ready data assumptions:
  - Y axis uses numeric ID (1..&MAXID) with an ID label format (IDLBL.)
  - X variables are day-based (e.g., TRT_X, DOSE_X, BOR_X, STATUS_X, DTH_X)
  - Dosing, response, status, prior therapy category variables are present
    and harmonized with the discrete attribute maps below.
=====*/
proc template;

```

```

/* Canvas for the entire figure */
define statgraph swim / store=work.templat;
  begingraph / designheight=9.5in designwidth=5in border=false;

/*=====
Section A. Discrete attribute maps for categorical encodings
- STATUS : treatment/participant status events on timelines
- DOSE   : dose types/arms (incl. crossover)
- OVRLRESP: overall response symbols along the timeline
- BOR    : best overall response symbols
=====*/

  (Discrete attribute section is omitted here.)

/*=====
Section B. Two-row lattice
Row 1 : main timelines panel
Row 2 : legend row (5 columns)
=====*/
layout lattice / columns=1 rows=2 rowweights= (0.9 0.1)
  rowdatarange=union border=false;

/*-----
(1) MAIN TIMELINE PANEL
- Y: Participant ID, labeled by IDLBL. format
- X: Study Day, includes pre-study to post-baseline window
- Layers:
  * SERIESPLOT for exposure (TRT_X) and concomitant meds (CM_X)
  * SCATTERPLOT for dose, OVRLRESP, prior therapy start/end, BOR,
  treatment status, and death
  * Reference line at Day=1 for alignment to first dose
-----*/
layout overlay / border=false
  yaxisopts=(type=linear label="Participant ID"
    griddisplay=on display=(ticks tickvalues)
    linearopts=(tickvalueformat=idlbl. viewmin=0
      viewmax=%eval(&maxid +1)
      tickvaluesequence=(start=1 end=&maxid increment=1)))
  xaxisopts=(type=linear label="Study Day"
    griddisplay=on 1
    inearopts=(viewmin=-500 viewmax=273
      tickvaluesequence=(start=-476 end=&maxdy increment=14)));
/* Exposure timeline by participant (solid black line) */
  seriesplot x=trt_x y=id / group=subjid_x grouporder=ascending
    includemissinggroup=false
    lineattrs=(pattern=solid thickness=1 color=black);
/* Concomitant medication interval (dashed black line) */
  seriesplot x=cm_x y=id / group=cmgrp includemissinggroup=false
    lineattrs=(pattern=dash thickness=1 color=black);
/* Dose events (by dose type/arm) */
  scatterplot x=dose_x y=id / group=dose name="dose";
/* On-timeline overall response assessments */
  scatterplot x=ovrlresp_x y=id / group=ovrlresp name="ovrlresp";
/* Prior anti-cancer therapy markers: start (▶) and end (◀) */
  scatterplot x=cmstdy_x y=id / group=cmst markerattrs=(symbol=TriangleRightFilled
    color=steelblue size=6pt) name="lactst";
  scatterplot x=cmendy_x y=id / group=cmen markerattrs=(symbol=TriangleLeftFilled
    color=steelblue size=6pt) name="lacten";
/* Arrows to indicate if the CM is too early and exceed the minview of x axis */
  scatterplot x=cm_x y=id / markerattrs=(symbol=arrowleft color=steelblue
    size=6pt) ;
/* BOR occurrence prior to study treatment */
  scatterplot x=bor_x y=id / group=bor name="bor";
/* Treatment/participant status events (AE, PD, Other, Ongoing) */
  scatterplot x=status_x y=id / group=status name="status";
/* Death marker (X) with larger size for visibility */
  scatterplot x=dth_x y=id / group=dth markerattrs=(symbol=X color=black size=9pt)
    name="dth";

```

```

/* Vertical reference line around Day 1 (first dose) */
referenceline x=1 / lineattrs=(pattern=MediumDash thickness=1 color=black);

endlayout;

/*-----
(2) LEGEND ROW - 5 columns
- Keeps the main panel uncluttered; each overlay hosts one legend
-----*/
layout lattice / columns=5 columnweights=(0.2 0.2 0.2 0.2 0.2)
rowdatarange=union border=false;

(The legend section is omitted here.)

endlayout; /* end legend row lattice */
endlayout; /* end 2-row main lattice */
endgraph;
end;
run;

```

Program 2. GTL template for Swimmer Plot of Dosing, Objective Response and Treatment Status

SWIMMER LANE VISUALIZATION OF TREATMENT EXPOSURE AND CLINICAL EVENTS (FIGURE 5 AND FIGURE 6)

Developed to provide a detailed, participant-level view of treatment exposure over time. These visualizations incorporate dosing schedules for each treatment component, adverse events (AEs), laboratory tests and relevant concomitant medication usage, allowing assessment of potential drug-drug interactions or mitigation strategies.




Figure 5. Swimmer Lanes for Grade 2 or above Adverse Events




Figure 6. Swimmer Lanes for Grade 2 or above Adverse Event with Lab Test

A SAS GTL **STATGRAPH** template creates a two row lattice:

- The top panel combines a baseline lab **AXISTABLE** and a timeline overlay with “Participant ID” and “Study Day” axes. **DISCRETEATTRMAP** defines markers for treatment status, dose, TEAE grade (with color-coded lines), and lab categories. The timeline uses **SERIESPLOT** for treatment and AE intervals, and **SCATTERPLOT/TEXTPLOT** for doses, AE start/end, labs, status, and end-of-study.
- The bottom row arranges legends for labs, TEAE grade, treatment, status, and participant status for clarity.
- The layouts are similar as the previous swimmer plot (refer to program 2).

MULTI-PANEL STACKED BAR CHART

CYCLE-BASED INCIDENCE ANALYSIS OF ADVERSE EVENTS (FIGURE 7)

To evaluate the temporal distribution of adverse events (AEs), horizontal bar plots were generated to display the incidence rate of AEs of interest by treatment cycle (Figure 7). This visualization highlights the percentage of patients experiencing AEs initiated within each cycle, enabling the study team to identify patterns in AE onset and assess potential cycle-specific risks. The cycle-level granularity supports informed safety monitoring and treatment planning.




Figure 7. Treatment-Emergent AEs of Interest by Cycle and Severity

We created a SAS GTL **TEMPLATE** using a **DATALATTICE** by *cycle* to compare AE severity across treatment groups.

A **DISCRETEATTRMAP** maps CTCAE grades (1–5) to distinct fill colors.

Within each lattice cell, a horizontal **BARCHART** stacks grade-specific proportions by treatment, with white outlines for separation, and a **SCATTERPLOT** prints the total number of participants to the right of each bar.

A rightaligned **SIDEBAR** hosts a **DISCRETELEGEND** for grade colors. Borders are suppressed to keep the focus on the distributions.

Patial GTL template code for Figure 7 is as below (Program 3):

```

/*=====
Program: Cycle-based Incidence of AEs of Interest by CTCAE Grade (GTL)
Template: STATGRAPH SWIMAE (stored in WORK.TEMPLAT)
Purpose :
  - Create a cycle-wise (row-wise) lattice.
  - Within each cycle, show % of participants with max CTCAE grade
    for an AE of interest by treatment group using stacked horizontal bars.
  - Print total % to the right of each bar.
  - Provide a right-aligned legend for CTCAE grades.
Plot-ready data assumptions:
  - CYCLE : cycle identifier used for row faceting (rowvar=cycle)
  - TRTN  : categorical treatment group (row axis within each cell)
  - ATOXGRN: CTCAE grade (1-5) used as the stacking group
  - PCT   : percentage of participants for that grade/treatment
  - PCTC  : total % across grades for that treatment (for markercharacter)
  - SUMC  : character to display as markercharacter (e.g., "N=39")
=====*/

```

```

proc template;
  /* Canvas for the entire figure */
  define statgraph swimae / store=work.templat;

  /*=====
  Section A. Grade color map: CTCAE grade → bar fill color
  - Ensure consistency across all lattice cells.
  =====*/
  (Discrete attribute section is omitted here.)
  /*=====
  Section B. DATALATTICE by cycle
  - One row per cycle with left row headers (cycle labels).
  - Row axis displays treatment groups; column axis displays % scale.
  - Slight gutter between rows for clarity.
  =====*/

  layout datalattice rowvar=cycle /
    headerlabeldisplay=value          /* show cycle label text */
    rowheaders=left                  /* row headers on the left side */
    headerborder=false
    headerlabelattrs=(weight=bold size=10pt)
    rowgutter=10pt border=false
    /* Left axis inside each cell: treatment categories */
    rowaxisopts=( display=(ticks tickvalues label)
    label="Treatment group" labelposition=top )
    /* Bottom axis inside each cell: % scale for horizontal bars */
    columnaxisopts=(display=(ticks tickvalues line label)
    label="Participants with Maximum CTCAE Grade AE of Interest (%)"
    linearopts=(viewmax=102) );

  /*-----
  PROTOTYPE layout: reused for each cycle cell
  - Horizontal stacked bars by grade
  - White outlines separate stacked segments
  - Right-justified total % printed via markercharacter
  -----*/

  layout prototype / walldisplay=none ;
    /* Stacked horizontal bars by grade within each treatment group */
    barchart x=trtn y=pct / group=atoxgrn groupdisplay=stack
    outlineattrs=(color=white) name='atox' orient=horizontal barwidth=1 ;
    /* Print total % to the right end of each treatment bar */
    scatterplot x=pctc y=trtn / markercharacter=sumc
    markercharacterattrs=(weight=bold size=9pt) markercharacterposition=right;
  endlayout; /* end prototype */

  /*-----
  Right-aligned sidebar legend (single column)
  -----*/

  sidebar / align=right ;
  layout overlay / border=false ;
    DiscreteLegend "atox" / title="Grade" valign=bottom across=1 border=false;
  endlayout;
  endsidebar;
  endlayout; /* end datalattice */

endgraph;
end;
run;

```

Program 3. GTL template for Treatment-Emergent AEs of Interest by Cycle and Severity

CONCLUSION

The integration of advanced SAS visual analytics has significantly enhanced the interpretation of complex clinical data. These tools have facilitated cross-functional collaboration, supported strategic decision-making, and helped translate patient-level insights into meaningful clinical impact. This approach demonstrates the value of tailored visualizations in accelerating oncology drug development.

ACKNOWLEDGMENTS

Thanks to the outstanding professionals at Gilead Sciences, Inc.

RECOMMENDED READING

- *Base SAS® Procedures Guide*
- *SAS® 9.4 Graph Template Language: User's Guide, Fifth Edition*

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Ceng Qian
Gilead Sciences, Inc.
ceng.qian1@gilead.com

Xu Wen
Gilead Sciences, Inc.
xu.wen2@gilead.com